

Quantile regression through linear programming

Anton Antonov

Mathematica for Prediction blog

Mathematica for Prediction project at GitHub

December 2013

Introduction

We can say that least squares linear regression corresponds to finding the mean of a single distribution. Similarly, quantile regression corresponds to finding quantiles of a single distribution. With quantile regression we obtain curves -- called "regression quantiles" -- that together with the least squares regression curve would give a more complete picture of the distributions (the y 's) corresponding to a set of x 's.

For a complete, interesting, and colorful introduction and justification to quantile regression see [2]. An introduction and description of the major properties of quantile regression is given in the Wikipedia entry [3].

In order to have a fast enough for practical purposes implementation of quantile regression we need to re-cast the quantile regression problem as linear programming problem. (Such a formulation is also discussed in [2].)

This document is mostly a guide for usage of the *Mathematica* package for quantile regression that is provided by the MathematicaForPrediction project at GitHub, see [1].

The second section provides theoretical background of the linear programming formulation of the quantile regression problem. The third section shows examples of finding regression quantiles using the function `QuantileRegressionFit` provided by [1]. The last section describes profiling experiments and their results.

The motivational examples in the theoretical section, formulas (1) and (2), can be completed with more expansions and proofs. (Which will be done in the next version of the document.)

Theory

We can formulate the quantile regression problem in way analogous to the formulation of least squares (conditional mean) regression.

Consider a random variable Y having some distribution function F and a sample $\{y_i\}_{i=1}^n$ of Y . The median of the set of samples $\{y_i\}_{i=1}^n$ can be defined as the solution of the minimization problem

$$\min_{\beta} \sum_{i=1}^n |y_i - \beta|, \quad \beta \in \mathbb{R}. \quad (1)$$

To see that the β which minimizes (1) is the median of $\{y_i\}_{i=1}^n$, consider two points $y_1 < y_2$. Then $|y_1 - \mu| + |\mu - y_2| = |y_1 - y_2|$, $\forall \mu \in [y_1, y_2]$, hence any $\mu \in [y_1, y_2]$ minimizes (1). Using the observation for two points we see that for three points $y_1 < y_2 < y_3$, $\mu = y_2$ minimizes (1). For four points $y_1 < y_2 < y_3 < y_4$ any $\mu \in [y_2, y_3]$ minimizes (1). We can generalize these observations and show that (1) gives the median for any set of points.

If we want to find θ -th sample quantile of $\{y_i\}_{i=1}^n$ then we need to change (1) into

$$\min_{\beta} \left(\sum_{y_i \geq \beta} \theta |y_i - \beta| + \sum_{y_i < \beta} (1 - \theta) |y_i - \beta| \right), \quad \beta \in \mathbb{R}. \quad (2)$$

Consider a set of random variables Y_i , $i \in [1, n]$, $n \in \mathbb{N}$ that are paired with a set of x -coordinates $X = \{x_i\}_{i=1}^n$. We have data of pairs $\{x_i, y_i\}_{i=1}^n$, where y_i is a realization of Y_i .

The linear regression problem can be formulated as

$$\min_{\beta_0, \beta_1} \sum_{i=1}^n (y_i - (\beta_0 + \beta_1 x_i))^2. \quad (3)$$

Similarly, the median regression problem can be formulated as

$$\min_{\beta_0, \beta_1} \sum_{i=1}^n |y_i - (\beta_0 + \beta_1 x_i)|. \quad (4)$$

and the θ -th quantile regression problem as

$$\min_{\beta_0, \beta_1} \left(\sum_{i \in \{i: y_i \geq \beta_0 + \beta_1 x_i\}} \theta |y_i - (\beta_0 + \beta_1 x_i)| + \sum_{i \in \{i: y_i < \beta_0 + \beta_1 x_i\}} (1 - \theta) |y_i - (\beta_0 + \beta_1 x_i)| \right), \quad \beta_0, \beta_1 \in \mathbb{R}. \quad (5)$$

In order to convert (5) into a linear programming problem, let us introduce the non-negative variables u_i and v_i for which the following equations are true:

$$\begin{aligned} y_i - (\beta_0 + \beta_1 x_i) + u_i &= 0, \quad i \in \{i: y_i \geq \beta_0 + \beta_1 x_i\}, \\ u_i &= 0, \quad i \notin \{i: y_i \geq \beta_0 + \beta_1 x_i\}, \end{aligned} \quad (6)$$

$$\begin{aligned} (\beta_0 + \beta_1 x_i) - y_i + v_i &= 0, \quad i \in \{i: y_i < \beta_0 + \beta_1 x_i\}, \\ v_i &= 0, \quad i \notin \{i: y_i < \beta_0 + \beta_1 x_i\}. \end{aligned} \quad (7)$$

Since u_i and v_i are greater than 0 on complementary sets, we can re-write (6) and (7) simply as

$$y_i - (\beta_0 + \beta_1 x_i) + u_i - v_i = 0, \quad u_i \geq 0, \quad v_i \geq 0, \quad i \in [1, n]. \quad (8)$$

Then (5) expressed with u_i and v_i becomes

$$\min_{u_i, v_i, \beta_0, \beta_1} \left(\sum_{i \in \{i: y_i \geq \beta_0 + \beta_1 x_i\}} \theta u_i + \sum_{i \in \{i: y_i < \beta_0 + \beta_1 x_i\}} (1 - \theta) v_i \right). \quad (9)$$

Since $u_i \geq 0$ and $v_i \geq 0$ the minimization function (9) can simply be written as

$$\min_{u_i, v_i, \beta_0, \beta_1} \left(\sum_{i=1}^n \theta u_i + \sum_{i=1}^n (1 - \theta) v_i \right). \quad (10)$$

The equations (8) and formula (10) are the linear programming formulation of the quantile regression problem (5).

Note that $u_i v_i = 0, \forall i \in [1, n]$.

The quantile regression formulations (5), and (8) and (10) can be done for any model of Y_i that is a linear combination of functions over X not just for the linear model $\beta_0 + \beta_1 X$.

Examples of usage

Package load


Load the package [1]:

```
In[1]:= Get["~/MathFiles/MathematicaForPrediction/QuantileRegression.m"]
```

Logarithmic curve with noise

Let us generate some data.

```
In[2]:= Block[{n = 1200, start = 10, end = 200},
  data = Table[{t, 5 + Log[t] +
    RandomReal[SkewNormalDistribution[0, Log[t] / 5, 12]]},
    {t, Rescale[Range[1, n], {1, n}, {start, end}]}];
ListPlot[data, AxesLabel → {"x", "y"},
  PlotRange → All, ImageSize → 400]
```



Consider the following quantiles:

```
In[4]:= qs = {0.05, 0.25, 0.5, 0.75, 0.95};
```

We want to find curves that separate the data according the quantiles. Those curves are called “regression quantiles”.

Pretending that we do not know how the data is generated, just by looking at the plot we assume that the model for the data is

$$y = \beta_0 + \beta_1 x + \beta_2 \sqrt{x} + \beta_3 \log(x). \quad (11)$$

Let us put the model functions for the regression fit in the variable `funcs`:

```
In[5]:= funcs = {1, x, Sqrt[x], Log[x]};
```

Here we find the regression quantiles:

```
In[6]:= qrFuncs = QuantileRegressionFit[data, funcs, x, qs];
TableForm[List /@ qrFuncs]
```

Out[7]//TableForm=


```
5.03476 + 0.00913189 Sqrt[x] + 2.15707 × 10-15 x + 0.98281 Log[x]
4.98702 + 4.08128 × 10-11 Sqrt[x] + 1.69792 × 10-13 x + 1.0654 Log[x]
5.07619 + 2.84134 × 10-12 Sqrt[x] + 2.09895 × 10-11 x + 1.11739 Log[x]
5.09638 + 1.02751 × 10-12 Sqrt[x] + 1.04385 × 10-12 x + 1.21172 Log[x]
5.48266 + 2.24466 × 10-14 Sqrt[x] + 4.18194 × 10-14 x + 1.30597 Log[x]
```

We also apply `Fit` to the data and the model functions in order to compare the regression quantiles with the least-squares regression fit:

```
In[8]:= fFunc = Fit[data, funcs, x]
```

```
Out[8]= 5.53539 + 0.349617 Sqrt[x] - 0.0111451 x + 0.53445 Log[x]
```

Here is a plot that combines the found regression quantiles and least squares fit:



Let us check how good the regression quantiles are for separating the data according to the quantiles they were computed for:

```

In[12]:= tbl = Table[
  {qs[[i]], Length[Select[data, #[[2]] ≥ (qrFuncs[[i]] /. x → #[[1]]) &]] /
    Length[data] // N}, {i, Length[qs]}}];
TableForm[tbl, TableHeadings → {None,
  {"quantile", "fraction\nabove"}}]

```

Out[13]//TableForm=

quantile	fraction above
0.05	0.950833
0.25	0.749167
0.5	0.499167
0.75	0.249167
0.95	0.0491667


Robustness

Let us demonstrate the robustness of the regression quantiles with the data of the previous example. Suppose that for some reason all the data y -values greater than 11.25 are altered by multiplying them with a some greater than 1 factor, say, $\alpha = 1.2$. Then the altered data looks like this:

```

In[14]:= α = 1.2;
dataAlt = Map[If[#[[2]] > 11.25, {#[[1]], α#[[2]]}, #] &, data];
ListPlot[dataAlt, AxesLabel → {"x", "y"},
  PlotRange → All, ImageSize → 400]

```



Let us compute the regression quantiles for the altered data:

```
In[17]:= qrFuncsAlt = QuantileRegressionFit[dataAlt, funcs, x, qs];
TableForm[List /@ qrFuncs]
```

Out[18]//TableForm=

$$5.03476 + 0.00913189 \sqrt{x} + 2.15707 \times 10^{-15} x + 0.98281 \log[x]$$

$$4.98702 + 4.08128 \times 10^{-11} \sqrt{x} + 1.69792 \times 10^{-13} x + 1.0654 \log[x]$$

$$5.07619 + 2.84134 \times 10^{-12} \sqrt{x} + 2.09895 \times 10^{-11} x + 1.11739 \log[x]$$

$$5.09638 + 1.02751 \times 10^{-12} \sqrt{x} + 1.04385 \times 10^{-12} x + 1.21172 \log[x]$$


$$5.48266 + 2.24466 \times 10^{-14} \sqrt{x} + 4.18194 \times 10^{-14} x + 1.30597 \log[x]$$

and let us also compute the least squares fit of the model (11):

```
In[19]:= fFuncAlt = Fit[dataAlt, funcs, x]
```

Out[19]= $6.36794 + 0.529118 \sqrt{x} - 0.00904845 x - 0.0328413 \log[x]$

Here is a plot that combines the functions found over the altered data:



We can see that the new regression quantiles computed for 0.05, 0.25, and 0.5 have not changed significantly:

```
In[23]:= qrFuncs[[1 ;; 3]]
```

Out[23]= $\left\{ \begin{aligned} &5.03476 + 0.00913189 \sqrt{x} + 2.15707 \times 10^{-15} x + 0.98281 \log[x], \\ &4.98702 + 4.08128 \times 10^{-11} \sqrt{x} + 1.69792 \times 10^{-13} x + 1.0654 \log[x], \\ &5.07619 + 2.84134 \times 10^{-12} \sqrt{x} + 2.09895 \times 10^{-11} x + 1.11739 \log[x] \end{aligned} \right\}$

```
In[24]:= qrFuncsAlt[[1 ;; 3]]
```

Out[24]= $\left\{ \begin{aligned} &5.03476 + 0.00913189 \sqrt{x} + 3.87407 \times 10^{-15} x + 0.98281 \log[x], \\ &4.98702 + 8.04435 \times 10^{-10} \sqrt{x} + 5.84834 \times 10^{-12} x + 1.0654 \log[x], \\ &5.07619 + 2.49087 \times 10^{-13} \sqrt{x} + 1.25704 \times 10^{-12} x + 1.11739 \log[x] \end{aligned} \right\}$

ant that they are still good for separating the un-altered data:

```
In[25]:= tbl = Table[
  {qs[[i]], Length[Select[data, #[[2]] ≥ (qrFuncsAlt[[i]] /. x → #[[1]]) &]] /
    Length[data] // N}, {i, Length[qs]}; TableForm[tbl,
  TableHeadings → {None, {"quantile", "fraction\nabove"}}]
```

Out[25]//TableForm=

quantile	fraction above
0.05	0.950833
0.25	0.750833
0.5	0.499167
0.75	0.204167
0.95	0.0116667

Also we can see that the least squares fit of (11) has significantly changed:

```
In[26]:= fFunc
```


```
Out[26]= 5.53539 + 0.349617  $\sqrt{x}$  - 0.0111451 x + 0.53445 Log[x]
```

```
In[27]:= fFuncAlt
```

```
Out[27]= 6.36794 + 0.529118  $\sqrt{x}$  - 0.00904845 x - 0.0328413 Log[x]
```

Data generated with Sin and noise

```
In[94]:= With[{n = 2000, start = 10, end = 140},
  dataSN = Table[{t,
    Sin[10 + t  $\pi$  / 15] + RandomReal[NormalDistribution[0, Log[t] / 3]]},
    {t, Rescale[Range[1, n], {1, n}, {start, end}]}];
  dataSN[[All, 2]] = dataSN[[All, 2]] + Log[dataSN[[All, 1]]];
]
ListPlot[dataSN, PlotRange → All, ImageSize → 400]
```



We are going to use again the quantiles:

```
In[51]:= qs = {0.05, 0.25, 0.5, 0.75, 0.95};
```

We can derive a guess about the data model by observing that we have peaks at $x = 50$, $x = 80$, $x = 110$. With these observations we make equations for Solve and take the second solution:

```
In[52]:= Solve[{a > 0, Sin[x α] == Sin[α (x + 60)], Sin[x α] == Sin[α (x + 30)]}, α]
```

```
Out[52]:= Solve[{a > 0, Sin[1.2 x] == Sin[1.2 (60 + x)],  
Sin[1.2 x] == Sin[1.2 (30 + x)]}, 1.2]
```

```
In[53]:= π / 15 // N
```

```
Out[53]:= 0.20944
```

Next we need to find a guess for the phase. Again we use the second solution provided by Solve:

```
In[54]:= Solve[{φ > 0, Sin[φ + 50 π / 15] == 1}, φ]
```

```
Out[54]:= {{φ → ConditionalExpression[  
1/6 (-5 π - 12 π C[1]), C[1] ∈ Integers && C[1] ≤ -1]},  
{φ → ConditionalExpression[1/6 (7 π - 12 π C[1]),  
C[1] ∈ Integers && C[1] ≤ 0]}}
```

```
In[55]:= 1/6 7 π // N
```


```
Out[55]:= 3.66519
```

Alternatively, we can simply use Manipulate and plot the data together with a model function subject to different parameters change.

```

In[96]:= Manipulate[
  DynamicModule[{gr1, gr2},
    gr1 = ListPlot[dataSN, PlotRange -> All];
    gr2 = Plot[ $\alpha \sin[\phi + \beta x]$ , {x, 0, 140}, PlotStyle -> Darker[Red]];
    Show[{gr1, gr2}]
  ], {{ $\alpha$ , 1}, 0.5, 10, 1}, {{ $\beta$ , 0, 2, 0.01}, { $\phi$ , 0, 30, 0.25]}

```



From the calculations we did so far we assume that the model for the data is

$$y = \beta_0 + \beta_1 x + \beta_2 \sin[3.7 + x \pi / 15]$$

Let us put the model functions for the regression fit in the variable `funcs`:

```

In[57]:= funcs = {1, x, Sin[3.7 + x  $\pi$  / 15]};

```

We find the regression quantiles:

```
In[97]:= qrFuncs = QuantileRegressionFit[dataSN, funcs, x, qs];
Grid[List /@ qrFuncs]
```


```
1.23145 + 0.00851515 x + 1.02497 Sin[3.7 +  $\frac{\pi x}{15}$ ]
2.19726 + 0.0139957 x + 1.02081 Sin[3.7 +  $\frac{\pi x}{15}$ ]
Out[98]= 2.79455 + 0.0187037 x + 1.04776 Sin[3.7 +  $\frac{\pi x}{15}$ ]
3.58857 + 0.020715 x + 1.05949 Sin[3.7 +  $\frac{\pi x}{15}$ ]
4.6321 + 0.0255846 x + 0.958566 Sin[3.7 +  $\frac{\pi x}{15}$ ]
```

As in the previous example we also apply `Fit` to the data and the model functions in order to compare the regression quantiles with the least-squares regression fit:

```
In[99]:= fFunc = Fit[dataSN, funcs, x]
```

```
Out[99]= 2.94185 + 0.0167552 x + 1.01842 Sin[3.7 +  $\frac{\pi x}{15}$ ]
```

Here is a plot that combines the functions found:



Let us check how good the regression quantiles are:

```

In[108]:= tbl = Table[
  {qs[[i]], Length[Select[dataSN, #[[2]] ≥ (qrFuncs[[i]] /. x → #[[1]]) &]] /
    Length[dataSN] // N}, {i, Length[qs]};
TableForm[tbl, TableHeadings → {None,
  {"quantile", "fraction\nabove"}}]

```

Out[109]//TableForm=

quantile	fraction above
0.05	0.95
0.25	0.7505
0.5	0.5005
0.75	0.2495
0.95	0.0505

Profiling

It is interesting to see the timing profile of the computations with `QuantileRegressionFit` across two axes: (i) data size and (ii) number of functions to be fit.

First we need to choose a family or several families of test data. Also, since *Mathematica*'s function `LinearProgramming` has several methods it is a good idea to test with all of them. Here I am going to show results only with one family of data and two `LinearProgramming` methods. The data family is the skewed noise over a logarithmic curve used as an example above. The first `LinearProgramming` method is *Mathematica*'s (default) "InteriorPoint", the second method is "CLP" that uses the built-in COIN-OR CLP optimizer. I run the profiling tests using one quantile {0.5} and five quantiles {0.05, 0.25, 0.5, 0.75, 0.95}, which are shown in blue and red respectively. I also run tests with different number of model functions $\{1, x, \sqrt{x}, \text{Log}[x]\}$ and $\{1, x, \text{Log}[x]\}$ but there was no significant difference in the timings (less than 2%).

Test family functions definitions

In this sub-section are shown definitions of functions generating families of data sets.

```
Clear[LogarithmicCurveWithNoise]
LogarithmicCurveWithNoise[
  nPoints_Integer, start_?NumberQ, end_?NumberQ] :=
Block[{data},
  data = Table[{t, 5 + Log[t] +
    RandomReal[SkewNormalDistribution[0, Log[t] / 5, 12]]},
    {t, Rescale[Range[1, nPoints], {1, nPoints}, {start, end}]}];
  data
];

Clear[SinWithUpwardTrend]
SinWithUpwardTrend[
  nPoints_Integer, start_?NumberQ, end_?NumberQ] :=
Block[{data},
  data = Table[{t,
    Sin[t  $\pi$  / 15] + RandomReal[NormalDistribution[0, Log[t] / 3]]},
    {t, Rescale[Range[1, nPoints], {1, nPoints}, {start, end}]}];
  data[[All, 2]] = data[[All, 2]] + data[[All, 1]]^1 / 6;
  data
];
```

```

Clear[SinWithParabolaTrend]
SinWithParabolaTrend[
  nPoints_Integer, start_?NumberQ, end_?NumberQ] :=
Block[{data},
  data = Table[
    {t, Sin[t  $\pi$  / 10] + RandomReal[NormalDistribution[0,  $\sqrt{t}$  / 10]]},
    {t, Rescale[Range[1, nPoints], {1, nPoints}, {start, end}]}];
  data[[All, 2]] = data[[All, 2]] +
    (Mean[{start, end}] - data[[All, 1]])^2 / 300;
  data
];

```

LogarithmicCurveWithNoise 4 model functions with Method→ LinearProgramming

```

modelFuncs = {1, x,  $\sqrt{x}$ , Log[x]};

qs = {0.05`, 0.25`, 0.5`, 0.75`, 0.95`}
{0.05, 0.25, 0.5, 0.75, 0.95}


dataSets = LogarithmicCurveWithNoise[#, 10., 100.] & /@
  Range[500, 10 000, 500];
dataSets // Length
20

timingsLogarithmicCurveWithNoiseF4Q1 = Map[
  {Length[#], AbsoluteTiming[QuantileRegressionFit[#, modelFuncs,
    x, {0.5}, Method→LinearProgramming];][[1]]} &, dataSets]
{{500, 0.098310}, {1000, 0.298962},
 {1500, 0.588580}, {2000, 0.987977}, {2500, 1.412103},
 {3000, 1.961045}, {3500, 2.658844}, {4000, 3.302903},
 {4500, 3.824936}, {5000, 4.812682}, {5500, 5.670178},
 {6000, 7.059653}, {6500, 8.519677}, {7000, 9.950919},
 {7500, 10.886583}, {8000, 12.807758}, {8500, 14.694095},
 {9000, 16.205932}, {9500, 17.815996}, {10 000, 21.027593}}

```

```
timingsLogarithmicCurveWithNoiseF4Q5 = Map[
  {Length[#], AbsoluteTiming[QuantileRegressionFit[#, modelFuncs,
    x, qs, Method -> LinearProgramming];] [[1]]} &, dataSets]
{{500, 0.549529}, {1000, 1.494088},
 {1500, 2.591068}, {2000, 3.493207}, {2500, 5.204162},
 {3000, 6.551883}, {3500, 7.290538}, {4000, 7.828214},
 {4500, 11.299679}, {5000, 12.629508}, {5500, 13.832353},
 {6000, 18.640458}, {6500, 19.818611}, {7000, 21.795818},
 {7500, 23.999628}, {8000, 27.048659}, {8500, 30.410464},
 {9000, 33.484113}, {9500, 38.420541}, {10000, 38.569162}}
```

```
ListLogPlot[{timingsLogarithmicCurveWithNoiseF4Q1,
  timingsLogarithmicCurveWithNoiseF4Q5},
  PlotStyle -> {PointSize[0.012]},
  PlotLegends -> SwatchLegend[{Darker[Blue], Darker[Red]},
    {"one quantile", "five quantiles"}],
  AxesLabel -> Map[Style[#, Larger] &, {"data size", "time,s"}],
  PlotLabel -> Style[
    "QuantileRegressionFit[___,Method->LinearProgramming]\ntimings
    per data size with four model functions\nfor
    skewed noise over a logarithmic curve",
    Larger], PlotRange -> All, ImageSize -> 600]
QuantileRegressionFit[___,Method->LinearProgramming]
timings per data size with four model functions
for skewed noise over a logarithmic curve
```




Average ratio between the execution times.

```
Mean[timingsLogarithmicCurveWithNoiseF4Q1[[All, 2]] /
      timingsLogarithmicCurveWithNoiseF4Q5[[All, 2]]]
0.380121
```

Ratios between the execution times:

```
ListPlot[Transpose[{timingsLogarithmicCurveWithNoiseF4Q1[[All, 1]],
                    timingsLogarithmicCurveWithNoiseF4Q1[[All, 2]] /
                    timingsLogarithmicCurveWithNoiseF4Q5[[All, 2]]}],
          PlotStyle -> {PointSize[0.02]}, PlotRange -> All]
```



LogarithmicCurveWithNoise 4 model functions with Method→ {LinearProgramming,Method→"CLP"}

The same model functions and quantiles were used as in the previous sub-section.

```
timingsLogarithmicCurveWithNoiseCLPF4Q1 =
  Map[{Length[#], AbsoluteTiming[
      QuantileRegressionFit[#, modelFuncs, x, {0.5}, Method ->
        {LinearProgramming, Method -> "CLP"}];][[1]] &, dataSets]

{{500, 0.057988}, {1000, 0.196983},
 {1500, 0.419909}, {2000, 0.795889}, {2500, 1.187998},
 {3000, 1.706534}, {3500, 2.287483}, {4000, 2.910575},
 {4500, 3.583797}, {5000, 4.425360}, {5500, 5.352640},
 {6000, 6.446963}, {6500, 8.402463}, {7000, 9.487401},
 {7500, 10.838176}, {8000, 12.300428}, {8500, 14.839014},
 {9000, 15.998385}, {9500, 17.698214}, {10000, 20.404420}}
```




```

timingsLogarithmicCurveWithNoiseCLPF4Q5 =
  Map[{Length[#], AbsoluteTiming[
    QuantileRegressionFit[#, modelFuncs, x, qs, Method →
      {LinearProgramming, Method → "CLP"}];] [[1]] &, dataSets]
{{500, 0.120797}, {1000, 0.411864},
 {1500, 0.913235}, {2000, 1.595715}, {2500, 2.369918},
 {3000, 3.433828}, {3500, 4.666453}, {4000, 6.038684},
 {4500, 7.456060}, {5000, 9.125450}, {5500, 10.848903},
 {6000, 13.103446}, {6500, 15.463412}, {7000, 17.807215},
 {7500, 20.780654}, {8000, 23.596392}, {8500, 26.951688},
 {9000, 31.034202}, {9500, 34.052749}, {10000, 38.549676}}

ListLogPlot[{timingsLogarithmicCurveWithNoiseCLPF4Q1,
  timingsLogarithmicCurveWithNoiseCLPF4Q5},
  PlotStyle → {PointSize[0.012]},
  PlotLegends → SwatchLegend[{Darker[Blue], Darker[Red]},
    {"one quantile", "five quantiles"}],
  AxesLabel → Map[Style[#, Larger] &, {"data size", "time,s"}],
  PlotLabel → Style[
    "QuantileRegressionFit[__, Method → {LinearProgramming, Method → \"
      CLP\"]]\ntimings per data size with four model
      functions\nfor skewed noise over a logarithmic curve",
    Larger], PlotRange → All, ImageSize → 600]

QuantileRegressionFit[__, Method → {LinearProgramming, Method → "CLP"}]
  timings per data size with four model functions
  for skewed noise over a logarithmic curve

```




Average ratio between the execution times.

```
Mean[timingsLogarithmicCurveWithNoiseCLPF4Q1[All, 2] /
      timingsLogarithmicCurveWithNoiseCLPF4Q5[All, 2]]
0.50362
```

Ratios between the execution times:

```
ListPlot[
  Transpose[{timingsLogarithmicCurveWithNoiseCLPF4Q1[All, 1],
             timingsLogarithmicCurveWithNoiseCLPF4Q1[All, 2] /
             timingsLogarithmicCurveWithNoiseCLPF4Q5[All, 2]}],
  PlotStyle -> {PointSize[0.02]}, PlotRange -> All]
```



Notes

It is interesting to note that the average ratio of the timings with 1 vs. 5 quantiles is 0.38 for "InteriorPoint" and 0.5 for "CLP".

Careful with "CLP"

During the profiling experiments with some of data families was observed that "CLP" gives curves that do not follow the data closely. Below is shown such a computation with both "InteriorPoint" and "CLP", the former looks good, the latter does not.

```
In[66]:= modelFuncs =
      {1, x, Sin[x π / 10], -Sin[x π / 10], -Sin[x π / 200], Sin[x π / 200]};

In[67]:= qs = {0.05`, 0.25`, 0.5`, 0.75`, 0.95`}
Out[67]= {0.05, 0.25, 0.5, 0.75, 0.95}

In[68]:= dataSet = SinWithParabolaTrend[1500, 10., 100.];


In[69]:= AbsoluteTiming[
      qrFuncs = QuantileRegressionFit[dataSet,
        modelFuncs, x, qs, Method -> {LinearProgramming}]
]

Out[69]= {4.828194, {7.73641 + 0.353175 x - 37.4138 Sin[ $\frac{\pi x}{200}$ ] + 1.09561 Sin[ $\frac{\pi x}{10}$ ],
      8.11931 + 0.358496 x - 37.2938 Sin[ $\frac{\pi x}{200}$ ] + 1.07894 Sin[ $\frac{\pi x}{10}$ ],
      8.42616 + 0.361935 x - 37.2894 Sin[ $\frac{\pi x}{200}$ ] + 1.01131 Sin[ $\frac{\pi x}{10}$ ],
      8.69552 + 0.364615 x - 37.1227 Sin[ $\frac{\pi x}{200}$ ] + 1.03673 Sin[ $\frac{\pi x}{10}$ ],
      9.02949 + 0.366971 x - 36.8279 Sin[ $\frac{\pi x}{200}$ ] + 1.01898 Sin[ $\frac{\pi x}{10}$ ]]}}
```

```

In[70]:= grData = ListPlot[dataSet, PlotRange → All];
grFit = Plot[Evaluate[MapThread[Tooltip[#1, #2] &, {qrFuncs, qs}]],
  {x, Min[dataSet[[All, 1]], Max[dataSet[[All, 1]]]},
  PlotStyle → AbsoluteThickness[1.4]];
Show[{grData, grFit}, ImageSize → 500]

```



```

In[72]:= AbsoluteTiming[
  qrFuncs = QuantileRegressionFit[dataSet, modelFuncs, x, qs, Method →
    {LinearProgramming, Method → "CLP", Tolerance → 10^-14.0}]
]

```

Out[72]=

$$\left\{ 0.852718, \left\{ -2.6634 + 2.62856 \sin\left[\frac{\pi x}{200}\right] + 0.998518 \sin\left[\frac{\pi x}{10}\right], \right. \right.$$

$$2.71231 - 2.98137 \sin\left[\frac{\pi x}{200}\right] + 0.898953 \sin\left[\frac{\pi x}{10}\right],$$

$$-2.6634 + 5.42374 \sin\left[\frac{\pi x}{200}\right] + 0.597093 \sin\left[\frac{\pi x}{10}\right],$$


$$-2.6634 + 7.41221 \sin\left[\frac{\pi x}{200}\right] - 0.0484257 \sin\left[\frac{\pi x}{10}\right],$$

$$\left. \left. -2.6634 + 21.3234 \sin\left[\frac{\pi x}{200}\right] - 1.80728 \sin\left[\frac{\pi x}{10}\right] \right\} \right\}$$

```

In[73]:= grData = ListPlot[dataSet, PlotRange → All];
grFit = Plot[Evaluate[MapThread[Tooltip[#1, #2] &, {qrFuncs, qs}]],
  {x, Min[dataSet[[All, 1]], Max[dataSet[[All, 1]]]},
  PlotStyle → AbsoluteThickness[1.4]];
Show[{grData, grFit}, ImageSize → 500]

```



References

- [1] Anton Antonov, Quantile regression *Mathematica* package, source code at GitHub, <https://github.com/antononcube/MathematicaForPrediction>, package QuantileRegression.m, (2013).
- [2] Roger Koenker, Gilbert Bassett Jr., “Regression Quantiles”, *Econometrica*, 46(1), 1978, pp. 33-50.
JSTOR URL: <http://links.jstor.org/sici?sici=0012-9682%28197801%2946%3A1%3C33%3AARQ%3E2.0.CO%3B2-J>.
- [3] Wikipedia, Quantile regression, http://en.wikipedia.org/wiki/Quantile_regression.
- [4] Brian Cade, Barry Noon, “A gentle introduction to quantile regression for ecologists”, *Front. Ecol. Environ.* 1(8), 2003, pp. 412–420.